


***=====Curve_Tracer_Tests=====**

Sometimes both simulations and real silicon will give out the exact same unexpected result. This is what happened while building the following LapTop curve-tracer.


http://www.idea2ic.com/LapTop_CurveTracer/LaptopCurve/BiCMOS_LapTop_CurveTracer.html

While working with Charles Williams, who is the developer of MacSpice, it was discovered that plotting X vs Y plots in MacSpice did not behave the same as they did on an oscilloscope. There was this strange blanking of the X/Y curve. This meant that one needed to view 2 dimensional relationships by using only plotting points.




=====
This blanking was removed quite a while ago to allow the X/Y plots to look much better.

Graph 43 - tran21: X-Y Plot example



Just recently, it became apparent why the line blanking feature was there in the first place. Now an old simulation of an NPN on the curve tracer no longer produces the same result.


NPN_CurveTracer



```

=====Setup=====
VCE C 0 DC 2V
IB 0 B DC 10u
Q1 C B 0 NPN1
=====Run_and_Plot=====
.control
=====Vsweep=Start=Stop==StepBy=Isweep=Start=Stop==StepBy=====
dc vce 0V 5V 0.1V ib 1uA 10uA 2uA
set pensize = 2
plot -i(vce) vs c


```



The laptop curve tracer circuit produces the exact same result if the collector voltage is seeing a sawtooth. It turns out that curve tracers don't sweep collector voltages with a sawtooth, but rather with half wave sine waves. The laptop curve tracer needed to be modified to use a triangle wave.

But spice creates something like a **sweep** voltage which does not have the flyback. This **sweep** voltage is set to be the default x axis for sweep plots. Simply plotting only collector current produces the correct results.

plot **-i(vce)**



The **sweep** is actually the collector voltage from the spice statement.

```
*=====Vsweep=Start=Stop==StepBy=Isweep=Start=Stop==StepBy=====
dc      vce  0V  5V  0.1V  ib  1uA  10uA  2uA
```





The **display** statement shows what waveforms get generated during an NPN curve-tracer sweep.


```
display
```

```
b          : voltage, real, 306 long
c          : voltage, real, 306 long
sweep     : voltage, real, 306 long [default scale]
vce#branch : current, real, 306 long
```


If one wants to take a better look at all the waveforms, one needs to construct an **index** array to be applied to the X direction.

```
*****Look_At_Array*****
let   num = length(sweep)
compose index start = 1 stop = $&num step = 1
plot  b      vs index
plot  c      vs index
plot  sweep  vs index
plot  vce#branch vs index
plot  -i(vce) vs index
```





NPN_CurveTracer



But as Charles suggested, perhaps the most correct thing to do is to assign the collector voltage as the default x scale.

```


setscale c
display
plot -i(vce)

```

```

b          : voltage, real, 306 long
c          : voltage, real, 306 long [default scale]
index     : notype, real, 306 long
num       : notype, real, 1 long
sweep    : voltage, real, 306 long
vce#branch : current, real, 306 long


```



This way the x axis gets correctly label as the collector voltage node.

=====Full_Netlist_For_Copy_Paste=====

NPN_CurveTracer



```

*
*
*=====Setup=====
VCE      C      0      DC      2V
IB       0      B      DC      10u
Q1       C      B      0      NPN1
*=====Run_and_Plot=====
.control
*=====Vsweep=Start=Stop==StepBy=Isweep=Start=Stop==StepBy=====
dc       vce    0V    5V    0.1V  ib    1uA  10uA  2uA
*Plot=====Collector_I=====
set      pensize = 2
plot     -i(vce)
display

*=====Look_At_Array=====
let      num = length(sweep)
compose index start = 1 stop = $&num step =1

plot     b          vs index
plot     c          vs index
plot     sweep      vs index
plot     vce#branch vs index
plot     -i(vce)    vs index

plot     -i(vce)    vs c

setscale c
display
plot     -i(vce)

.endc
.model  NPN1          NPN( BF=70 VAF=60 )
.END
*plot   -i(vce)      vs    c    pointplot

```

7.12.10_10.31AM
dsauersanjose@aol.com
Don Sauer
<http://www.idea2ic.com/>